Some Useful Maple Commands

d i f f: Compute symbolic derivatives

```
> ex1 := exp( sin(4*x) ) + ln( x^2 + sqrt(x) );
```

$$ex1 := e^{\sin(4x)} + \ln\left(x^2 + \sqrt{x}\right)$$

```
> diff( ex1, x );
```

$$4\cos(4x)\, e^{\sin(4x)} + \frac{2x + \frac{1}{2\sqrt{x}}}{x^2 + \sqrt{x}}$$

Higher order derivatives

```
> diff( ex1, x, x );
```

$$-16\sin(4x)\, e^{\sin(4x)} + 16\cos(4x)^2\, e^{\sin(4x)} + \frac{2 - \frac{1}{4x^{3/2}}}{x^2 + \sqrt{x}} - \frac{\left(2x + \frac{1}{2\sqrt{x}}\right)^2}{\left(x^2 + \sqrt{x}\right)^2}$$

equivalently ...

```
> diff( ex1, x$2 );
```

$$-16\sin(4x)\, e^{\sin(4x)} + 16\cos(4x)^2\, e^{\sin(4x)} + \frac{2 - \frac{1}{4x^{3/2}}}{x^2 + \sqrt{x}} - \frac{\left(2x + \frac{1}{2\sqrt{x}}\right)^2}{\left(x^2 + \sqrt{x}\right)^2}$$

Multivariate differentiation

Note nesting of **d i f f** commands

```
> diff( diff( cos(x/y), x ), y );
```

$$\frac{\cos\left(\frac{x}{y}\right) x}{y^3} + \frac{\sin\left(\frac{x}{y}\right)}{y^2}$$

i n t: Compute symbolic integrals (anti-derivatives and definite integrals)

```
> ex2 := 1 / sqrt( x^2 - a^2 );
```

$$ex2 := \frac{1}{\sqrt{-a^2 + x^2}}$$

```
> int( ex2 , x );
```

$$\ln\left(x+\sqrt{-a^2+x^2}\right)$$

Observe that the output does not include an arbitrary constant of integration

Definite integrals

```
> int( x^2 , x = 0 .. 2 );
```

$$\frac{8}{3}$$

Don't expect miracles!! I.e. many expressions have no definite integral expressible in terms of elementary functions. in which case the output from Maple is identical to the input

```
> int( ex1, x );
```

$$\int \left(e^{\sin(4x)}+\ln\left(x^2+\sqrt{x}\right)\right)\,dx$$

Multi-dimensional integrals are often straightforward ... note nesting of **int** commands

```
> ex3 := (x^3  + y^3) / (x^2 - y^2);
```

$$ex3 := \frac{x^3+y^3}{x^2-y^2}$$

```
> int( int(ex3, x), y);
```

$$\frac{1}{6}x^2y-\frac{1}{3}(x-y)^3\ln(x-y)-\frac{1}{9}y^3-\frac{1}{6}y^2x+\frac{11}{18}x^3+xy^2\ln(x-y)-\ln(x-y)\,yx^2$$

series and taylor: Compute power series expansions

```
> series( exp(x) , x = 0);
```

$$1+x+\frac{1}{2}x^2+\frac{1}{6}x^3+\frac{1}{24}x^4+\frac{1}{120}x^5+O(x^6)$$

Optional third argument specifies expansion order

```
> series( exp(x), x = 0, 11);
```

$$1+x+\frac{1}{2}x^2+\frac{1}{6}x^3+\frac{1}{24}x^4+\frac{1}{120}x^5+\frac{1}{720}x^6+\frac{1}{5040}x^7+\frac{1}{40320}x^8+\frac{1}{362880}x^9+\frac{1}{3628800}x^{10}+O(x^{11})$$

For purposes of course, **taylor** and **series** are synonymous commands

```
> taylor( exp(x), x = 0);
```

$$1 + x + \frac{1}{2} x^2 + \frac{1}{6} x^3 + \frac{1}{24} x^4 + \frac{1}{120} x^5 + O(x^6)$$

The special Maple variable **Order** controls the default expansion order

```
> Order := 8;
```

$$Order := 8$$

```
> taylor(exp(x), x = 0);
```

$$1 + x + \frac{1}{2} x^2 + \frac{1}{6} x^3 + \frac{1}{24} x^4 + \frac{1}{120} x^5 + \frac{1}{720} x^6 + \frac{1}{5040} x^7 + O(x^8)$$

```
> Order := 6;
```

$$Order := 6$$

```
> taylor_exp := taylor(exp(x), x = 0);
```

$$taylor\_exp := 1 + x + \frac{1}{2} x^2 + \frac{1}{6} x^3 + \frac{1}{24} x^4 + \frac{1}{120} x^5 + O(x^6)$$

Convert the last Taylor series to a polynomial (useful for Homework 2!)

```
> taylor_exp_p := convert(taylor_exp, polynom);
```

$$taylor\_exp\_p := 1 + x + \frac{1}{2} x^2 + \frac{1}{6} x^3 + \frac{1}{24} x^4 + \frac{1}{120} x^5$$

Evaluate the series approximation (polynomial) for x = 0.2

```
> res1 := evalf( subs(x = 0.2, taylor_exp_p) );
```

$$res1 := 1.221402667$$

Compute the corresponding exact value as a floating point number

```
> res2 := exp(0.2);
```

$$res2 := 1.221402758$$

Compute the error in the approximation

```
> err := res1 - res2;
```

$$err := -9.1\,10^{-8}$$

solve: Solve equations, including linear systems

```
> eq1 := x + y + z = 6; eq2 := 2 * x + y + z = 2;
  eq3 := x + y + 3 * z = 3;
```

$$eq1 := x + y + z = 6$$
$$eq2 := 2\,x + y + z = 2$$
$$eq3 := x + y + 3\,z = 3$$

```
> solve( {eq1, eq2, eq3} , {x, y, z} );
```

$$\left\{x = -4,\ y = \frac{23}{2},\ z = -\frac{3}{2}\right\}$$

Since there are only three distinct names in the set of equations, Maple can deduce that they are the three unknowns ...

```
> solve( {eq1, eq2, eq3} );
```

$$\left\{x = -4,\ y = \frac{23}{2},\ z = -\frac{3}{2}\right\}$$

Common mistakes

(1) Not terminating statement with ';' or ':'

```
> a := int(x^2, x = 0..1)
```

Warning, inserted missing semicolon at end of statement

$$a := \frac{1}{3}$$

(2) Using '=' rather than ':=' for assignment ... the following evaluates as an equation (whose logical value is false)

```
> a = 2;
```

$$\frac{1}{3} = 2$$

The variable **a** was previously assigned a value, which was not altered by the above statement

```
> a;
```

$$\frac{1}{3}$$

This is what we meant to type

```
> a := 2;
```

$$a := 2$$

```
> a;
```

$$2$$

(3) Using single quotes (') rather than double quotes (") to delimit strings

```
> string1 := "This is a Maple string";
```

$$string1 := \text{"This is a Maple string"}$$

```
> string2 := 'This is not a Maple string';
Error, missing operator or `;`
```